

# Challenge January 2024

## Family Riddle

A solution with OPL CPLEX by Alex Fleischer  
afleischer@fr.ibm.com

OPL (Optimization Programming Language) is an algebraic modeling language that helps model easily optimization problems that can be solved both with IBM CPLEX linear programming and IBM CPLEX constraint programming CPOptimizer (CPO)

NB: You can use free CPLEX Community Edition for this.

You get

```
In the family René
boys are 1 3 8 girls are 0 5 7
In the family Léo
boys are 4 5 9 girls are 3 7 8
nb solutions=1
```

With the OPL CPLEX CPOptimizer model

```
using CP;

//Let's assume that Rene and Leo are both heads of household, and, what a
coincidence,
//both families include three girls and three boys.

{string} families={"René","Léo"};
{string} gender={"boy","girl"};
range r=1..3;

// All the children are under age ten.
dvar int+ age[families][gender][r] in 0..9;

subject to
{
// Break sym
// Neither family includes any twins, nor any children closer in age than a year.

forall(f in families, g in gender,i in r:i!=1) age[f][g][i-1]+1<=age[f][g][i];
forall(f in families,i in r,j in r) age[f]["boy"][i]!=age[f]["girl"][j];

// The youngest child in Leo's family is a girl, and in Rene's family, a little
girl has just arrived.
age["Léo"]["girl"][1]+1<=age["Léo"]["boy"][1];
age["René"]["girl"][1]==0;
```

```

// In each family, the sum of the ages of the girls is equal to the sum of the
ages of the boys
forall(f in families) sum(i in r) age[f]["boy"][i]==sum(i in r) age[f]["girl"][i];

// the sum of the squares of the ages of the girls is equal to the sum of the
squares of the ages of the boys.
forall(f in families) sum(i in r) age[f]["boy"][i]*age[f]["boy"][i]==
sum(i in r) age[f]["girl"][i]*age[f]["girl"][i];

// The sum of the ages of all these children is 60.
sum(f in families,g in gender,i in r) age[f][g][i]==60;
}

execute
{
  for(var f in families)
  {
    writeln("In the family ",f);
    write("boys are ");
    for(var i in r) write(age[f]["boy"][i]," ");
    write("girls are ");
    for(var i in r) write(age[f]["girl"][i]," ");
    writeln();
  }
}

// Loop to enumerate all solutions
main
{
  var nbsol=0;
  cp.param.SearchType=24;
  cp.param.workers=1;

  thisOplModel.generate();
  cp.startNewSearch();
  while
  (cp.next()) { thisOplModel.postProcess(); nbsol++; }
  writeln("nb solutions=",nbsol);
}

```